We will need to solve some linear equations, so we load the library.

```
> with(linalg):
```

```
Warning, new definition for norm
Warning, new definition for trace
```

Much of what Maple does with vectors is more convenient using their "list" data structure, so we use that until vectors or matrices are needed.

```
> xlist:=[0,1/4,1/2,3/4,1];
```

$$xlist := \left[0, \frac{1}{4}, \frac{1}{2}, \frac{3}{4}, 1\right]$$

```
> ylist:=[seq(cos(Pi*xlist[i]),i=1..5)];
```

$$ylist := \left[1, \frac{1}{2}\sqrt{2}, 0, -\frac{1}{2}\sqrt{2}, -1\right]$$

Using a separate analysis, we found that the equations defining the derivatives at these points for the free spline satisfy Ax=b for A and b defined below.

```
> A:=matrix([[2,1,0,0,0],[1,4,1,0,0],[0,1,4,1,0],[0,0,1,4,1],[0,0,0,
  1,2]]);
```

$$A := \begin{bmatrix} 2 & 1 & 0 & 0 & 0 \\ 1 & 4 & 1 & 0 & 0 \\ 0 & 1 & 4 & 1 & 0 \\ 0 & 0 & 1 & 4 & 1 \\ 0 & 0 & 0 & 1 & 2 \end{bmatrix}$$

```
> b:=evalm(6*vector([sqrt(2)-2,-2,-2*sqrt(2),-2,sqrt(2)-2]));
```

$$b := [6\sqrt{2} - 12, -12, -12\sqrt{2}, -12, 6\sqrt{2} - 12]$$

```
> y1list:=convert(linsolve(A,b),list);
```

$$y1list := [-5 + 3\sqrt{2}, -2, 1 - 3\sqrt{2}, -2, -5 + 3\sqrt{2}]$$

These are the derivatives at the values in xlist. This is all that we need to find the Hermite cubics on each interval. We do this by building the lists of all divided differences.

```
> fa:=ylist[1..4];
```

$$fa := \left[1, \frac{1}{2}\sqrt{2}, 0, -\frac{1}{2}\sqrt{2}\right]$$

```
> fb:=ylist[2..5];
```

$$fb := \left[\frac{1}{2}\sqrt{2}, 0, -\frac{1}{2}\sqrt{2}, -1\right]$$

```
> fab:=4*(fb-fa);
```

$$fab := [2\sqrt{2} - 4, -2\sqrt{2}, -2\sqrt{2}, 2\sqrt{2} - 4]$$

Here, we are using the fact that all subintervals are of length 1/4.

```
> faa:=y1list[1..4];
```

$$faa := [-5 + 3\sqrt{2}, -2, 1 - 3\sqrt{2}, -2]$$

```
> fbb:=y1list[2..5];
```

$$fbb := [-2, 1-3\sqrt{2}, -2, -5+3\sqrt{2}]$$

```
> faab:=4*(fab-faa);
```

$$faab := [4-4\sqrt{2}, 8-8\sqrt{2}, -4+4\sqrt{2}, -8+8\sqrt{2}]$$

```
> fabb:=4*(fbb-faa);
```

$$fabb := [12-12\sqrt{2}, 12-12\sqrt{2}, -12+12\sqrt{2}, -12+12\sqrt{2}]$$

```
> faabb:=4*(fabb-faab);
```

$$faabb := [32-32\sqrt{2}, 16-16\sqrt{2}, -32+32\sqrt{2}, -16+16\sqrt{2}]$$

```
> spline:=seq(simplify(fa[i]+(x-xlist[i])*(faa[i]+(x-xlist[i])*(faab
  [i]+(x-xlist[i+1])*faabb[i]))),i=1..4);
```

$$spline := 1-5x+3x\sqrt{2}-4x^2+4x^2\sqrt{2}+32x^3-32x^3\sqrt{2},$$
$$\frac{1}{2}\sqrt{2}-x+16x^3-16x^3\sqrt{2}-8x^2+8x^2\sqrt{2}+\frac{1}{2}-x\sqrt{2},$$
$$\frac{1}{2}(2x-1)(-9+7\sqrt{2}+36x-36x\sqrt{2}-32x^2+32x^2\sqrt{2}),$$
$$-5\sqrt{2}-23x+32x^2-32x^2\sqrt{2}-16x^3+16x^3\sqrt{2}+21x\sqrt{2}+6$$

Normally, we would not simplify these expressions since they were constructed in a form that is better for computation. In this case, however, we know all values exactly, and the interval is fairly short, so we do not meet any of the usual causes of error in this formula.

```
> fsp:=x->piecewise(x<xlist[2],spline[1],x<xlist[3],spline[2],x<xlis
  t[4],spline[3],spline[4]);
```

$$fsp := x \to \text{piecewise}(x < xlist_2, spline_1, x < xlist_3, spline_2, x < xlist_4, spline_3, spline_4)$$

```
> plot({cos(Pi*x),fsp(x)},x=0..1);
```


1
0.5
0
-0.5
-1
0.2
0.4
0.6
0.8
1
x

>